PROC HPFOREST
Parola chiave:  HPFOREST
Contesto: [DEFINIZIONE DELLA PROCEDURA] PROC HPFOREST

Syntax: PROC HPFOREST < option(s) > ;
    INPUT variable(s) < / option(s) > ; 
    TARGET variable < / LEVEL=level > ; 
    FREQ variable ;
    PERFORMANCE performance-options ;
    
The HPFOREST procedure is a high-performance procedure that creates a predictive model called a forest
that consists of several decision trees. A predictive model defines a relationship between input variables and
a target variable. The purpose of a predictive model is to predict a target value from inputs. The HPFOREST
procedure trains the model; that is, it creates the model, using training data in which the target values are
known. The model can then be applied to observations in which the target is unknown. If the predictions
fit the new data well, the model is said to generalize well. Good generalization is the primary goal for
predictive tasks. A predictive model might fit the training data well but generalize poorly.

A decision tree is a type of predictive model that has been developed independently in the statistics and
artificial intelligence communities. The HPFOREST procedure creates a tree recursively. An input variable
is chosen and used to create a rule to split the data into two segments. The process is then repeated in each
segment, and then again in each new segment, and so on until some constraint is met. In the terminology of
the tree metaphor, the segments are nodes, the original data set is the root node, and the final unpartitioned
segments are leaves or terminal nodes. A node is an internal node if it is not a leaf. The data in a leaf
determine the estimates of the value of the target variable. These estimates are subsequently applied to
predict the target of a new observation assigned to the leaf.

The HPFOREST procedure creates decision trees that differ from each other in two ways. First, the training
data for a tree is a sample, without replacement, from the original training data of the forest. Second, the
input variables considered for splitting a node are randomly selected from all available inputs. Among these
variables, the HPFOREST procedure considers only a single variable when forming a splitting rule. The
chosen variable is the one that is most associated with the target.

Training a forest can require training hundreds of decision trees. The HPFOREST procedure can exploit
computer grids by training trees in parallel independently on different grid nodes.

data spambase;
      %let url=//archive.ics.uci.edu/ml/machine-learning-databases;
      infile "http:&url/spambase/spambase.data"
            device=url delimiter=',';
      input wf_make wf_adress wf_all wf_3d wf_our
            wf_over wf_remove wf_internet wf_order wf_mail
            wf_receive wf_will wf_people wf_report wf_addresses
            wf_free wf_business wf_email wf_you wf_credit
            wf_your wf_font wf_000 wf_money wf_hp
            wf_hpl wf_george wf_650 wf_lab wf_labs
            wf_telnet wf_857 wf_data wf_415 wf_85
            wf_technology wf_1999 wf_parts wf_pm wf_direct
            wf_cs wf_meeting wf_original wf_project wf_re
            wf_edu wf_table wf_conference
            cf_semicolon cf_parenthese cf_bracket cf_exclamation
            cf_dollar cf_pound
            average longest total spam;
   run;
proc hpforest data=spambase maxtrees=200;
   input w: c: average longest total/level=interval;
   target spam/level=binary;
   ods output FitStatistics=fitstats(rename=(Ntrees=Trees));
run;

data fitstats;
   set fitstats;
   label Trees = 'Number of Trees';
   label MiscAll = 'Full Data';
   label Miscoob = 'OOB';
run;

proc sgplot data=fitstats;
   title "OOB vs Training";
   series x=Trees y=MiscAll;
   series x=Trees y=MiscOob/lineattrs=(pattern=shortdash thickness=2);
   yaxis label='Misclassification Rate';
run;
title;

Figure 16: Plot of OOB versus Training Misclassification Rate
[image: Plot of OOB versus Training Misclassification Rate]

Figure 16 shows the misclassification rate is worse (larger) based on the out-of-bag (OOB) data, and more trees are needed for the out-of-bag rates to level off. Both characteristics are typical of a forest.
Example 7.2 Number of Variables to Try When Splitting a Node
This example illustrates the effect of changing the number of variables to randomly select as candidate splitting variables in a node. In each node in each tree, m variables are randomly selected to be candidates to split on. Use the VARS_TO_TRY= option to specify m. Specifying m less than the number of available inputs is one way to reduce the correlation between the trees in the forest. Broadly speaking, the predictions of a forest improve when the trees are less correlated. On the other hand, the predictions of the forest improve when the predictions of the trees improve (without changing the correlations). When the number of useful inputs are much less than the total number of inputs, smaller values of m produce weaker trees because fewer nodes consider useful inputs for defining a splitting rule. Try several values of m to find a good one for the data.
The following SAS statements create a SAS data set from an URL:

   data spambase;
      %let url=//archive.ics.uci.edu/ml/machine-learning-databases;
      infile "http:&url/spambase/spambase.data"
            device=url delimiter=',';
      input wf_make wf_adress wf_all wf_3d wf_our
            wf_over wf_remove wf_internet wf_order wf_mail
            wf_receive wf_will wf_people wf_report wf_addresses
            wf_free wf_business wf_email wf_you wf_credit
            wf_your wf_font wf_000 wf_money wf_hp
            wf_hpl wf_george wf_650 wf_lab wf_labs
            wf_telnet wf_857 wf_data wf_415 wf_85
            wf_technology wf_1999 wf_parts wf_pm wf_direct
            wf_cs wf_meeting wf_original wf_project wf_re
            wf_edu wf_table wf_conference
            cf_semicolon cf_parenthese cf_bracket cf_exclamation
            cf_dollar cf_pound
            average longest total spam;
   run;
%macro hpforest(Vars=);
proc hpforest data=spambase maxtrees=200
   vars_to_try=&Vars.;
   input w: c: average longest total/level=interval;
   target spam/level=binary;
   ods output
   FitStatistics = fitstats_vars&Vars.(rename=(Miscoob=VarsToTry&Vars.));
run;
%mend;

%hpforest(vars=all);
%hpforest(vars=40);
%hpforest(vars=26);
%hpforest(vars=7);
%hpforest(vars=2);

data fitstats;
   merge
   fitstats_varsall
   fitstats_vars40
   fitstats_vars26
   fitstats_vars7
   fitstats_vars2;
   rename Ntrees=Trees;
   label VarsToTryAll = "Vars=All";
   label VarsToTry40 = "Vars=40";
   label VarsToTry26 = "Vars=26";
   label VarsToTry7 = "Vars=7";
   label VarsToTry2 = "Vars=2";
run;

proc sgplot data=fitstats;
   title "Misclassification Rate for Various VarsToTry Values";
   series x=Trees y = VarsToTryAll/lineattrs=(Color=black);
   series x=Trees y=VarsToTry40/lineattrs=(Pattern=ShortDash Thickness=2);
   series x=Trees y=VarsToTry26/lineattrs=(Pattern=ShortDash Thickness=2);
   series x=Trees y=VarsToTry7/lineattrs=(Pattern=MediumDashDotDot Thickness=2);
   series x=Trees y=VarsToTry2/lineattrs=(Pattern=LongDash Thickness=2);
   yaxis label='OOB Misclassification Rate';
run;
title;

Figure 17: Effect of the VARS_TO_TRY= Option on the Misclassification Rate

Example 7.3 Fraction of Training Data to Train a Tree
This example illustrates the effect of changing the fraction of original training observations used to train an individual tree. Use the INBAGFRACTION= option to specify f. Specifying f less than 1 is one way to reduce the correlation between the trees in the forest.
The following SAS statements create a SAS data set from an URL:

   data spambase;
      %let url=//archive.ics.uci.edu/ml/machine-learning-databases;
      infile "http:&url/spambase/spambase.data"
            device=url delimiter=',';
      input wf_make wf_adress wf_all wf_3d wf_our
            wf_over wf_remove wf_internet wf_order wf_mail
            wf_receive wf_will wf_people wf_report wf_addresses
            wf_free wf_business wf_email wf_you wf_credit
            wf_your wf_font wf_000 wf_money wf_hp
            wf_hpl wf_george wf_650 wf_lab wf_labs
            wf_telnet wf_857 wf_data wf_415 wf_85
            wf_technology wf_1999 wf_parts wf_pm wf_direct
            wf_cs wf_meeting wf_original wf_project wf_re
            wf_edu wf_table wf_conference
            cf_semicolon cf_parenthese cf_bracket cf_exclamation
            cf_dollar cf_pound
            average longest total spam;
   run;
%macro hpforest(f=, output_suffix=);
proc hpforest data=spambase maxtrees=500 vars_to_try=26
   trainfraction=&f;
   input w: c: average longest total/level=interval;
   target spam/level=binary;
   ods output
   FitStatistics = fitstats_f&output_suffix.(rename=(Miscoob=fraction&output_suffix.));
run;
%mend;

%hpforest(f=0.8, output_suffix=08);
%hpforest(f=0.6, output_suffix=06);
%hpforest(f=0.4, output_suffix=04);

data fitstats;
   merge
   fitstats_f08
   fitstats_f06
   fitstats_f04;
   rename Ntrees=Trees;
   label fraction08 = "Fraction=0.8";
   label fraction06 = "Fraction=0.6";
   label fraction04 = "Fraction=0.4";
run;

proc sgplot data=fitstats;
   title "Misclassification Rate for Various Fractions of Training Data";
   series x=Trees y=fraction08/lineattrs=(Pattern=ShortDash Thickness=2);
   series x=Trees y=fraction06/lineattrs=(Pattern=MediumDashDotDot Thickness=2);
   series x=Trees y=fraction04/lineattrs=(Pattern=LongDash Thickness=2);
   yaxis label='OOB Misclassification Rate';
run;
title;
Figure 18: Effect of the INBAGFRACTION Option on the Misclassification Rate

Specifying a value of 7 or 26 for the VARS_TO_TRY= option results in a more accurate forest than would occur without random selection of variables (VARS_TO_TRY=ALL). Specifying VARS_TO_TRY=2 is no better than specifying VARS_TO_TRY=ALL. A good value for the VARS_TO_TRY= option depends on the data. In this example, the HPFOREST procedure uses a default value of , which worked well.


PROC HPFOREST Statement
· PROC HPFOREST <options>;
The PROC HPFOREST statement invokes the procedure. You can specify one or more of the following optional arguments.
DATA=<libref.>SAS-data-set
names the SAS data set to be used by PROC HPFOREST for training the model. The default is the most recently created data set.
If the data are already distributed, the procedure reads the data alongside the distributed database. See the section Processing Modes for the various execution modes and the section Alongside-the-Database Execution for the alongside-the-database model. Data from all the computer grid nodes are combined into a structure that is optimized for model training and redistributed to the nodes. The different nodes then proceed independently with identical data to create decision trees.
ALPHA=number
specifies a threshold p-value for the significance level of a test of association of a candidate variable with the target. If no association meets this threshold, the node is not split. The default value is 1.
BALANCE=YES | NO
specifies whether to modify the splitting criterion for a nominal target so that the number of observations in each target class are effectively equal. A weight is applied to the count of observations in a class. The weight is different in different nodes. You can specify the following values:
YES
modifies the splitting criterion so that the number of observations in each target class are effectively equal. Setting BALANCE=YES can improve prediction when the class sizes are very different.
NO
does not modify the splitting criterion.
By default, BALANCE=NO.
CATBINS=k
specifies the maximum number of categories of a nominal candidate variable to use in the association test. k refers only to the categories that are present in the training data in the node and that satisfy the MINCATSIZE= option. The categories are counted independently in each node. If more than k categories are present, then the least frequent categories are removed from the association test. Many infrequent categories can dilute a strong predictive ability of common categories. The search for a splitting rule uses all categories that satisfy the MINCATSIZE= options. The value of k must be a positive integer. The default value is 30.
EXHAUSTIVE=number
specifies the maximum number of splits to examine in a complete enumeration of all possible splits when the input variable is nominal and the target has more than two nominal categories. The exhaustive method of searching for a split examines all possible splits. If the number of possible splits is greater than number, then a heuristic search is done instead of an exhaustive search. The default value of number is 5,000.
GRIDCLASSSIZE=n
specifies the minimum number of observations of any value of a binary or nominal target to exist on a grid node when GRIDCOPY=MINIMAL is specified. Observations are copied between grid nodes as necessary to meet this minimum. This option is ignored unless PROC HPFOREST is running in distributed mode. The default value is 10,000.
GRIDCOPY=ALL | MINIMAL | NONE | TRAINING
specifies how many observations to copy between grid nodes when PROC HPFOREST runs in distributed mode. PROC HPFOREST creates a decision tree on a single grid node without using data on other nodes unless observations are copied from other nodes before any tree is created. Generally, the more observations on a node, the larger the tree can grow and the more accurate the forest. The disadvantage is time, because the core processors have more work to do. You can specify the following values:
ALL
copies all training and validation data.
MINIMAL
copies enough training data to achieve a minimum on each node. This option uses the GRIDCLASSSIZE= and GRIDNODESIZE= option values to determine the minimum number of observations required on each grid node.
NONE
does not copy any observations.
TRAINING
copies all training data.
By default, GRIDCOPY=MINIMAL.
GRIDNODESIZE=n
specifies the minimum number of observations to use on a grid node when PROC HPFOREST runs in distributed mode and GRIDCOPY=MINIMAL is specified. If the number of observations on a node is less than n, then observations are copied from other nodes to achieve n. The default value is 100,000.
IMPORTANCE=YES | NO
specifies whether to compute variable importance. You can specify the following values:
YES
computes loss reduction variable importance.
NO
does not computes loss reduction variable importance. If you save the model (by specifying the SAVE statement) and subsequently input it to PROC HP4SCORE, then PROC HP4SCORE cannot compute variable importance either.
By default, IMPORTANCE=YES.
INBAGFRACTION=f
TRAINFRACTION=f
specifies the fraction of training observations to train a tree with, where f can be any number greater than 0 and at most 1. Using less than all the available data often improves the generalization error. A different in-bag sample is taken for each tree. The default value of f is 0.6. PROC HPFOREST uses at least four observations in the in-bag data regardless of how small f is (assuming four observations exist). If an observation is available for training but is not an in-bag datum, then it is either out-of-bag or a pruning datum. If f is too small to accommodate the LEAFSIZE=, LEAFFRACTION=, and SPLITSIZE options then no tree is made. The INBAGN= option accepts an absolute number instead of a fraction to specify the same quantity. Specifying both the INBAGN= and INBAGFRACTION= options is an error.
INBAGN=n
TRAINN=n
specifies how many observations to use to train each tree. The observations are counted without regard to the variable specified in the FREQ statement. Using less than all the available data often improves the generalization error. A different in-bag sample is taken for each tree. n can be any positive integer. If n is greater than the number of observations in the data set specified in the DATA= option, then all the available data are used. n must be at least 3 and large enough to accommodate the values of the LEAFSIZE=, LEAFFRACTION=, and SPLITSIZE options. The default value is 0.6 times the number of available observations in DATA= data set. The INBAGFRACTION= option accepts a fraction instead of an absolute number to specify the same quantity as the INBAGN= option. Specifying both the INBAGN= and INBAGFRACTION= is an error.
INTERVALBINS=k
specifies the number of equally spaced bins into which variables are divided when the PRESELECT=BINNEDSEARCH algorithm is executed. The default value is 100.
LEAFFRACTION=f
specifies the smallest number of training observations that a new branch can have, expressed as the fraction of the number N of available observations in the DATA= data set. N might be less than the total number of observations in the data set because observations with a missing target value or non positive value of the variable specified in the FREQ statement are excluded from N. If you specify a number in the LEAFSIZE= option that implies a larger number than that specified in the LEAFFRACTION= option, f is ignored. The value f must be larger than 0 and less than 1. The default value is 0.00001.
LEAFSIZE=n
specifies the smallest number of training observations a new branch can have. If you specify a value for the LEAFFRACTION= option that implies a larger value than n, the LEAFSIZE= option is ignored. The default value is 1.
MAXDEPTH=d
specifies the maximum depth of a node in any tree that PROC HPFOREST creates. The depth of a node equals the number of splitting rules needed to define the node. The root node has depth 0. The children of the root have depth 1, the children of those children have depth 2, and so on. The smallest acceptable value of d is 1. The default value of d is 20 (implying a maximum of 1,048,576 leaves).
MAXTREES=n
specifies the number of trees in the forest. n is a positive integer. The number of trees in the resulting forest can be less than n when the HPFOREST procedure fails to split the training data for a tree. Up to two times n trees are attempted. If the procedure fails to split the training data for more than n trees, then less than n trees are created. The ALPHA=, LEAFSIZE=, and MINCATSIZE= options constrain the split search to form trees that are more likely to predict well using new data. Setting all of these options to 1 generally frees the search algorithm to find a split and train a tree, although the tree might not help the forest predict well. The default value of n is 100.
MINCATSIZE=n
specifies the minimum number of observations that a given nominal input category must have in order to use the category in a split search. Categorical values that appear in fewer than n observations are handled as if they were missing. The categories that occur in fewer than n observations are merged into the pseudo category for missing values for the purpose of finding a split. The policy for assigning such observations to a branch is the same as the policy for assigning missing values to a branch. The default value of n is 5.
MINUSEINSEARCH=n
specifies a threshold for utilizing missing values in the split search when MISSING=USEINSEARCH is specified as the missing value policy. If the number of observations in which the splitting variable has missing values in a node is greater than or equal to n, then PROC HPFOREST initiates the USEINSEARCH policy for missing values. See the section Handling Missing Values for a more complete explanation. The default value of n is 1.
MISSING=USEINSEARCH | BIGBRANCH
specifies how the training procedure handles an observation with missing values. If MISSING=USEINSEARCH and the number of training observations in the node is more than n, where n is the value of the MINUSEINSEARCH= option, then the missing value is used as a separate, legitimate value in the test of association and the split search. If MISSING=BIGBRANCH, observations with a missing value of the candidate variable are omitted from the test of association and split search in that node. A splitting rule will assign such an observation to the branch containing the most observations among those used in the split search. See the section Handling Missing Values for a more complete explanation. By default, MISSING=USEINSEARCH.
NODESIZE=n | ALL
specifies the number of training observations to use for association tests and split searches. NODESIZE=ALL requests to use all the observations. The acceptable range is from two to two billion on most machines. The default value of n is 100,000.
The procedure counts the number of training observations in a node without adjusting the number with the values of the variable specified in the FREQ statement. If the count is larger than n, then the split search for that node is based on a random sample of size n. For categorical targets, the sample uses as many observations with less frequent target values as possible. The calculations for the association measures and split worth adjust the category counts to the category proportions in the node before sampling.
PRESELECT=BINNEDSEARCH | LOH | HOTHORN
specifies the method for selecting the variable to split with. You can specify the following values:
BINNEDSEARCH
selects the variable that produces the rule that has the largest worth. This option searches for a splitting rule for each candidate variable. Values of an interval variable are binned before the search. This option preferentially selects nominal variables that have many categories. But if such variables do not exist or if all the variables have a similar number of categories, then this option potentially selects better variables than the other methods. This option usually takes longer because searching for a splitting rule is more complex. For more information, see the section Binned Search.
HOTHORN
selects the variable that produces the rule that has the highest association with the target. For more information, see the section Hothorn, Hornik, and Zeileis.
LOH
selects the variable that has the smallest p-value of a chi-square test of association in a contingency table. For more information, see the section Loh and GUIDE.
If no nominal variables have more than five categories, then PRESELECT=BINNEDSEARCH by default. Otherwise, if some variables are not nominal or if any nominal variable has five categories more than another nominal variable, then PRESELECT=LOH by default.
PRUNEFRACTION=g
specifies the fraction of training observations that are available for pruning a split. The value of g can be any number from 0 and to 1, although a number close to 1 would leave little to grow the tree. The default value of g is 0; the default action is to not prune.
PRUNETHRESHOLD=t
specifies the lower limit of allowable shrinkage when the distance of the child node from the parent is measured by the pruning data instead of the in-bag data. For more information, see the section Pruning. The value of t must be between 0 and 1. The default value of t is 0.1.
SCOREPROLE=DEFAULT | INBAG | OOB | VALID
specifies which observations are used for the prediction of a new observation in the SCORE statement. Predictions are based on average values of observations that have the same role in leaves. All the observations included in an average have a single role: in-bag, out-of-bag, or validation. The SCOREPROLE= option does not affect the "Fit Statistics" output table. The HP4SCORE procedure outputs the same predictions as the HPFOREST procedure.
You can specify the following values:
DEFAULT
selects the default value as if the option was not specified.
INBAG
outputs predictions that are based on in-bag observations.
OOB
outputs predictions that are based on out-of-bag observations. If out-of-bag observations are not present, then the in-bag observations in the leaf are used.
VALID
outputs predictions that are based on validation observations. If validation observations are not present in a leaf node, the out-of-bag observations in that leaf are used. If out-of-bag observations are not present, then the in-bag observations in the leaf are used.
By default, SCOREPROLE=INBAG.
SEED=n
specifies the seed for generating random numbers. The HPFOREST procedure uses random numbers to select training observations for each tree and to select candidate variables in each node to split on. n is a nonnegative integer. Set n to 0 to use the default value. The default value of the seed is 8,976,153.
SKIP_SEQ_ROWS=n
specifies the number of rows to skip in the "Fit Statistics" table in distributed mode. After every n trees that are trained on a grid node, the fit statistics on the node are updated, consolidated with statistics from other nodes, and eventually output in the "Fit Statistics" table. Each row in the table contains statistics for a specific number of trees in a forest. The table has gaps of up to n rows. The gap is smaller if fewer than n trees are made on a grid node, and no gap appears if only one tree is made on a node. The SKIP_SEQ_ROWS= option has no affect in single-machine mode. By default, SKIP_SEQ_ROWS=5.
SPLITSIZE=n
specifies the requisite number of training observations a node must have for the HPFOREST procedure to consider splitting it. By default on is twice the value of the LEAFSIZE= option (or n is the value implied by LEAFFRACTION= option if the procedure ignores the LEAFSIZE= option). The procedure counts the number of observations in a node without adjusting the number with the values of the variable specified in the FREQ statement when it interprets the value specified in the LEAFFRACTION=, LEAFSIZE=, MINCATSIZE=, and SPLITSIZE= options.
TRAINFRACTION=f
is the same as the INBAGFRACTION= option.
TRAINN=n
is the same as the INBAGN= option.
VARS_TO_TRY=m | ALL
specifies the number of input variables to consider splitting on in a node. m ranges from 1 to the number of input variables, v. The default value of m is . Specify VARS_TO_TRY=ALL to use all the inputs as candidates in a node.

Machine Learning: Running A Random Forest In SAS
In order to run a Random forest in SAS we have to use the PROC HPFOREST specifying the target variable and outlining weather the variables are Categorical or Qunatitative. For the purpose of this analysis we utilized a target (Repsone Variable) that is Categorical (Nominal in SAS language) as described in the code of the image below in the colors yellow and red:

After running the code we get a series of tables that will prvide the detail in the analysis. The model information, for instance let us know that there are 3 variables that were randomly selected to test each node or possible split (yellow) in each tree. Also we can see that the maximum number of trees that were run were 100 as underlined in blue. The Model info also tells us that the “Inbag Fraction” was set at 60% which is the default, making the OBB at a rate of 40%. Please note that the “Prune Fraction” is defaulted at “0″ becasue the closest it is set to “1″ then the lowest level of growth the tree will have. In other words is not to prune. 
HPFOREST automatically uses only the valid variables that have no missing records under any of the observations. However we can also see that out of the 213 countries in the study sample, 213 have been utilized. This is due to the fact I have already utilized a set of counties that only have no missing values. 

Next we can see that the Model generates a table with “Baseline Fit Statistics”. In the case of the data in this study we can see that the model identified a 38% of miss-classification or in other words a rate of 62% of accurate classification . Thsi means that the majority of the sample has been classified correctly in each of the randomly selected samples. 

 When analyzing the fitness of the forest in the table below we can see that the misclassification rate has reached its lowest point with the tree number 100.This suggests that when this model is used to be tested in the OOB sample the misclassification rate its only at 22%.

Lastly, we see that the SAS POC HPFOREST gives us a table with “Loss Reduction Variable Importance” . This table outlines the rank of importance of how each variable contributes to the predictability of the model. As you can see in the graph below, the variable of Alcohol has ranked highest.
Now the following will helsp us to understand how to read the table:
· Number of Rules: Tells us the # of splitting rules that use a variable
· Each data is computed twice:
· Gini: This is data computed in a “Bagged” stage
· Gini OOB: This is data computed in an “Out of Bag” stage 
· The Fit statistics tell us that the OOB data is less biassed, therefore the data is sorted by the OOB Gini Measure 
· The variables are listed as highly importance(top) and lowest importance(bottom) in terms of predicting higher than the norm suicide rates. 
· From the table below we can see that the variables that are more likely to be a good fit in a model to predict higher than the norm suicide rates is Alcohol Consumption, Employment, and Urban Rate. 
[image: https://media.licdn.com/dms/image/C5112AQHLkwJs7vUUDg/article-inline_image-shrink_400_744/0/1520212619811?e=1684972800&v=beta&t=Wvxd9LPM8btLEw5ZfKyu9UqZLO36HMJSlwBgme0q_24]
By conducting the exercise above we can see that Random forests are a data mining algorithm that can select important Explanatory variables that can be used to determine the outcome of a response (Target) variable whether categorical or quantitative. Also this exercise allows us to use a combination of categorical and quantitative variables. In sum, this forest lets us know which variables are important but not the relationship to one another.
Credits: Special Thanks to Ph.D. Lisa Dierker, Ph. D. Jennifer Rose, Wesleyan University and Coursera for the great level of theoretical and intellectual contribution that I have utilized to generating this blog post. 

